pragma solidity ^0.4.21;

// File: zeppelin-solidity/contracts/math/SafeMath.sol

/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {

  /**
  * @dev Multiplies two numbers, throws on overflow.
  */
  function mul(uint256 a, uint256 b) internal pure returns (uint256 c) {
    if (a == 0) {
      return 0;
    }
    c = a * b;
    assert(c / a == b);
    return c;
  }

  /**
  * @dev Integer division of two numbers, truncating the quotient.
  */
  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    // uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return a / b;
  }

  /**
  * @dev Subtracts two numbers, throws on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  /**
  * @dev Adds two numbers, throws on overflow.
  */
  function add(uint256 a, uint256 b) internal pure returns (uint256 c) {
    c = a + b;
    assert(c >= a);
    return c;
  }
}

// File: contracts/BonusStrategy.sol

contract BonusStrategy {
    using SafeMath for uint;

    uint public defaultAmount = 1*10**18;
    uint public limit = 300*1000*10**18; // 300.000  DCNT
    uint public currentAmount = 0;
    uint[] public startTimes;
    uint[] public endTimes;
    uint[] public amounts;

    constructor(
        uint[] _startTimes,
        uint[] _endTimes,
        uint[] _amounts
        ) public 
    {
        require(_startTimes.length == _endTimes.length && _endTimes.length == _amounts.length);
        startTimes = _startTimes;
        endTimes = _endTimes;
        amounts = _amounts;
    }

    function isStrategy() external pure returns (bool) {
        return true;
    }

    function getCurrentBonus() public view returns (uint bonus) {
        if (currentAmount >= limit) {
            currentAmount = currentAmount.add(defaultAmount);
            return defaultAmount;
        }
        for (uint8 i = 0; i < amounts.length; i++) {
            if (now >= startTimes[i] && now <= endTimes[i]) {
                bonus = amounts[i];
                currentAmount = currentAmount.add(bonus);
                return bonus;
            }
        }
        currentAmount = currentAmount.add(defaultAmount);
        return defaultAmount;
    }

}

// File: zeppelin-solidity/contracts/token/ERC20/ERC20Basic.sol

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  function totalSupply() public view returns (uint256);
  function balanceOf(address who) public view returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

// File: zeppelin-solidity/contracts/token/ERC20/BasicToken.sol

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  uint256 totalSupply_;

  /**
  * @dev total number of tokens in existence
  */
  function totalSupply() public view returns (uint256) {
    return totalSupply_;
  }

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    emit Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256) {
    return balances[_owner];
  }

}

// File: zeppelin-solidity/contracts/token/ERC20/BurnableToken.sol

/**
 * @title Burnable Token
 * @dev Token that can be irreversibly burned (destroyed).
 */
contract BurnableToken is BasicToken {

  event Burn(address indexed burner, uint256 value);

  /**
   * @dev Burns a specific amount of tokens.
   * @param _value The amount of token to be burned.
   */
  function burn(uint256 _value) public {
    _burn(msg.sender, _value);
  }

  function _burn(address _who, uint256 _value) internal {
    require(_value <= balances[_who]);
    // no need to require value <= totalSupply, since that would imply the
    // sender's balance is greater than the totalSupply, which *should* be an assertion failure

    balances[_who] = balances[_who].sub(_value);
    totalSupply_ = totalSupply_.sub(_value);
    emit Burn(_who, _value);
    emit Transfer(_who, address(0), _value);
  }
}

// File: zeppelin-solidity/contracts/token/ERC20/ERC20.sol

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public view returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: zeppelin-solidity/contracts/token/ERC20/StandardToken.sol

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    emit Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    emit Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public view returns (uint256) {
    return allowed[_owner][_spender];
  }

  /**
   * @dev Increase the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _addedValue The amount of tokens to increase the allowance by.
   */
  function increaseApproval(address _spender, uint _addedValue) public returns (bool) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  /**
   * @dev Decrease the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To decrement
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _subtractedValue The amount of tokens to decrease the allowance by.
   */
  function decreaseApproval(address _spender, uint _subtractedValue) public returns (bool) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}

// File: contracts/InfoBurnableToken.sol

contract InfoBurnableToken is BurnableToken, StandardToken {
    string message = "No sufficient funds";
    address public manager;

    event NoFunds(address _who, string _message);

    modifier onlyManager() {
        require(msg.sender == manager);
        _;
    }

    constructor(address _manager) public {
        require(address(_manager) != 0);
        manager = _manager;
    }

    function burn(uint256 _value) public {
        if (balances[msg.sender] < _value){
            emit NoFunds(msg.sender, message);
        }else {
            _burn(msg.sender, _value);
        }
    }

    function burnPassportToken(address _from, uint256 _value) onlyManager public returns (bool) {
        if (_value <= balances[_from]){
            _burn(_from, _value);
            return true;
        }
        emit NoFunds(_from, message);
        return false;
    }

    function transferManager(address _newManager) onlyManager public returns (bool) {
        require(address(_newManager) != 0);
        manager = _newManager;
        return true;
    }

}

// File: contracts/DecenturionToken.sol

contract DecenturionToken is InfoBurnableToken {
    using SafeMath for uint;

    string constant public name = "Decenturion Token";
    string constant public symbol = "DCNT";
    uint constant public decimals = 18;
    uint constant public deployerAmount = 20 * (10 ** 6) * (10 ** decimals); // 20 000 000 DCNT
    uint constant public managerAmount = 10 * (10 ** 6) * (10 ** decimals); // 10 000 000 DCNT

    constructor(address _manager) InfoBurnableToken(_manager) public {
        totalSupply_ = 30 * (10 ** 6) * (10 ** decimals); // 30 000 000 DCNT
        balances[msg.sender] = deployerAmount;
        balances[manager] = managerAmount;
    }

}

// File: zeppelin-solidity/contracts/ownership/Ownable.sol

/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() public {
    owner = msg.sender;
  }

  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }

  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) public onlyOwner {
    require(newOwner != address(0));
    emit OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

}

// File: zeppelin-solidity/contracts/lifecycle/Pausable.sol

/**
 * @title Pausable
 * @dev Base contract which allows children to implement an emergency stop mechanism.
 */
contract Pausable is Ownable {
  event Pause();
  event Unpause();

  bool public paused = false;


  /**
   * @dev Modifier to make a function callable only when the contract is not paused.
   */
  modifier whenNotPaused() {
    require(!paused);
    _;
  }

  /**
   * @dev Modifier to make a function callable only when the contract is paused.
   */
  modifier whenPaused() {
    require(paused);
    _;
  }

  /**
   * @dev called by the owner to pause, triggers stopped state
   */
  function pause() onlyOwner whenNotPaused public {
    paused = true;
    emit Pause();
  }

  /**
   * @dev called by the owner to unpause, returns to normal state
   */
  function unpause() onlyOwner whenPaused public {
    paused = false;
    emit Unpause();
  }
}

// File: contracts/TokenManager.sol

contract TokenManager is Ownable, Pausable {
    using SafeMath for uint;

    uint public totalRegistrationBonusAmount = 0;
    uint public emissionTime;
    uint public tokensForBurn = 1*10**18; 
    string public alreadyRegisteredMessage = "user already registered";
    string public nullBonusMessage = "registration bonus = 0";

    mapping (address=>bool) isRegistered;
    mapping (address=>bool) isTokensDistributed;

    BonusStrategy public bonusStrategy;
    DecenturionToken public token;

    event UserAlreadyRegistered(address user, string message);
    event TokensAlreadyDistributed(address user);

    constructor(BonusStrategy _bonusStrategy, uint _emissionTime) public {
        require(_bonusStrategy.isStrategy());
        require(_emissionTime > now);
        bonusStrategy = _bonusStrategy;
        emissionTime = _emissionTime;
    }

    function() public payable { }

    function setTokenAddress(DecenturionToken _token) onlyOwner public returns (bool){
        require(address(_token) != 0);
        token = _token;
        return true;
    }

    function distributeRegistrationBonus(address _recepient) onlyOwner whenNotPaused public returns (bool) {
        if (!isRegistered[_recepient]) {
            _distributeRegistrationBonus(_recepient);
            isRegistered[_recepient] = true;
        }else {
            emit UserAlreadyRegistered(_recepient, alreadyRegisteredMessage);
            return false;
        }
        return true;
    }

    function _distributeRegistrationBonus(address _recepient) internal returns (bool) {
        uint registrationBonus = bonusStrategy.getCurrentBonus();
        totalRegistrationBonusAmount = totalRegistrationBonusAmount.add(registrationBonus);  
        token.transfer(_recepient, registrationBonus);
        return true;
    }

    function distributeTokens(address _address, uint _totalEthBalance) onlyOwner whenNotPaused public returns (bool) {
        require(now >= emissionTime);
        if (isTokensDistributed[_address]){
            emit TokensAlreadyDistributed(_address);
            return false;
        }
        uint decimals = 10**18;
        uint precision = 10**3;
        uint balance = _address.balance;
        uint stake = balance.div(_totalEthBalance);
        uint total = token.managerAmount().sub(totalRegistrationBonusAmount);
        uint emission = stake.mul(total).div(decimals).mul(precision);
        token.transfer(_address, emission);
        isTokensDistributed[_address] = true;
        return true;
    }

    function setEmissionTime(uint _time) onlyOwner whenNotPaused public returns (bool) {
        require(now <= _time);
        emissionTime = _time;
        return true;
    }

    function register(address _who) onlyOwner whenNotPaused public returns (bool) {
        if (isRegistered[_who]) {
            emit UserAlreadyRegistered(_who, alreadyRegisteredMessage);
            return false;
        }
        _distributeRegistrationBonus(_who);

        bool isBurned = token.burnPassportToken(_who, tokensForBurn);
        if (isBurned) {
            isRegistered[_who] = true;
            return true;
        }else{
            return false;
        }
    }

    function sendTokensTo(address[] _users, uint _amount) onlyOwner whenNotPaused public {
        uint maxLength = 10;
        if (_users.length < maxLength) {
            maxLength = _users.length;
        }
        for (uint i = 0; i < maxLength; i++){
            token.transfer(_users[i], _amount);
        }
    }

    function substituteManagerContract(address _newManager) onlyOwner whenNotPaused public {
        token.transferManager(_newManager);
    }

}